
ВВЕДЕНИЕ. РЕШЕНИЕ ЗАДАЧ 

ДЛЯ ГИПЕРУПРУГИХ МАТЕРИАЛОВ В ANSYS. 

Пакет прикладных программ ANSYS является универсальным средством 

решения широкого класса задач механики и математической физики. В том 

числе, ANSYS содержит в своем составе эффективные средства решения задач 

деформирования гиперупругих материалов при различных нагрузках и режи-

мах нагружения. Нелинейная механика гиперупругих сред компактно и доста-

точно подробно изложена в первой части данного пособия [1]. Характерными 

особенностями деформирования гиперупругих материалов являются, во-

первых, большие деформации и, во-вторых – несжимаемость или почти несжи-

маемость этих материалов. Напомним, что при больших перемещениях и де-

формациях исходная конфигурация тела не совпадает с конечной, деформиро-

ванной конфигурацией, как это принято в линейной механике, где перемещения 

и деформации считаются малыми по сравнению с единицей. Для описания не-

линейного деформирования используются специальные представления тензо-

ров напряжений и деформаций [2]. Нелинейная связь перемещений и деформа-

ций, а также то, что конечная деформированная конфигурация является изна-

чально неизвестной, делает задачу деформирования нелинейной. Несжимае-

мость означает вовсе не то, что гиперупругое тело нельзя деформировать, а то, 

что в процессе деформирования не изменяется его объем, что также накладыва-

ет свои особенности. 

Какие же эффективные инструменты для решения задач гиперупругости 

имеются в ANSYS? Это модели гиперупругих материалов, конечные элемен-

ты, предназначенные для расчета несжимаемых или почти несжимаемых тел 

при больших перемещениях и деформациях и, наконец, эффективные алго-

ритмы, позволяющие решать задачи для различных случаев нагружения. Рас-

смотрим подробнее некоторые из представленных в ANSYS моделей гипе-

рупругих сред и конечных элементов. 

МОДЕЛИ ГИПЕРУПРУГИХ СРЕД В ANSYS 

В ANSYS представлены следующие модели изотропных гиперупругих 

сред [3]: 

 модель неогукова материала; 

 модель Муни-Ривлина; 

 полиномиальная модель; 

 модель Огдена; 

 модель Арруды-Бойса; 

 модель Гента; 

 модель Йо. 

Кроме того, существуют анизотропные модели, есть возможность созда-

вать пользовательские модели. Из всего многообразия мы рассмотрим неогуко-

ву модель и модель Муни-Ривлина. Приведем краткие сведения об этих моде-

лях. Более подробная информация содержится в первой части пособия [1]. 

Модель неогукова материала 

Упругий потенциал неогукова материала имеет вид: 

2
1 )1(

1
)3(

2
 J

d
I

G
W  

W  – потенциал энергии деформаций; 

1I  – первый инвариант изохорического тензора деформаций Грина; 

G  – модуль сдвига; 

d – материальный параметр несжимаемости, связанный с начальным объём-

ным модулем 
d

K
2

 ; 

Модель материала Муни-Ривлина 

Упругий потенциал двухконстантного материала Муни-Ривлина имеет 

вид: 

2
201110 )1(

1
)3()3(  J

d
ICICW  

W  – потенциал энергии деформаций; 

21, II  – первый и второй инварианты изохорического тензора деформаций 

Грина; 

0110 ,CC  – материальные константы материала; 

d – материальный параметр несжимаемости; 

Начальные значения объемного модуля и модуля сдвига связаны с кон-

стантами материала следующим образом: 

)(2;
2

0110 CCG
d

K  . 

Как видно, модель неогука сводится к модели Муни-Ривлина при 001 С . 

Часть потенциала с коэффициентами 0110 ,CC  определяют девиаторную 

часть деформаций и напряжений, а последнее слагаемое – шаровую часть. При 

нулевом значении d модель определяет несжимаемую среду. Ненулевые значе-

ния этого параметра соответствуют различной степени сжимаемости. 

Чтобы представить, как ведут себя гиперупругие материалы при одноос-

ном растяжении, был проведен численный эксперимент. Был зафиксирован 

начальный модуль сдвига G = 1 МПа. При этом значения коэффициентов 

0110 ,CC  выбирались так, чтобы их сумма равнялась 0.5 МПа. На рис. 1-2 пока-

заны графики зависимости истинного напряжения Коши   
  от удлинения   

  

при одноосном растяжении при значениях объемного модуля K = 2 МПа и 

K = 100 МПа. 

  

Рис. 1. Зависимость напряжения Коши   
   

от удлинения   
  при K = 2. 

Рис. 2. Зависимость напряжения Коши   
   

от удлинения   
  при K = 100. 

Увеличение коэффициента     при постоянном значении модуля сдвига 

приводит к увеличению абсолютной величины напряжения   
  при растяжении 

и уменьшению его при сжатии. Увеличение начального значения объемного 

модуля приводит к увеличению напряжения   
 

 при растяжении. Еще можно 

отметить, что при рассмотренных изменениях кривые не пересекаются в точке 

с координатами (1, 0), а касаются друг друга. 

КОНЕЧНЫЕ ЭЛЕМЕНТЫ ДЛЯ РАСЧЕТА 

ГИПЕРУПРУГИХ СРЕД В ANSYS 

Целый ряд конечных элементов, предназначенных для аппроксимации 

тел различной мерности, может работать с моделями гиперупругих сред. Спи-

сок этих элементов показан в таблице 1. 

Таблица 1. 

Основные конечные элементы для расчета гиперупругих сред 

Расчетные схемы Конечные элементы 

Пластины и оболочки SHELL181, SHELL281, SOLSH190 

Двумерные задачи PLANE182,  PLANE183 

Трехмерные задачи SOLID185, SOLID186 , SOLID187 

Далее мы рассмотрим, в основном, использование только конечных эле-

ментов для расчета трехмерных тел, а именно SOLID185. Использование дру-

гих типов элементов студенты могут рассмотреть самостоятельно. 

Элемент SOLID185 представляет собой криволинейный гексаэдр (шести-

гранник) с 8-ю узлами, при построении которого используется изопараметриче-

ская технология [4]. Известно, что такие элементы обладают существенными 

недостатками, такими как сдвиговое и объемное запирание или заклинивание. 

Эффект запирания состоит в том, что жесткость элемента при некоторых видах 

деформирования может необоснованно возрастать. Так, сдвиговое запирание 

возникает в задачах, где основным деформированным состоянием является из-

гиб. Объемное запирание появляется в задачах для материалов, близких к не-

сжимаемым. Причина этих эффектов заключается в использовании в качестве 

функций формы неполных полиномов. Разработаны различные подходы [4, 5, 

6] для преодоления этих негативных явлений. 

В ANSYS также присутствуют средства, позволяющие преодолеть нега-

тивные эффекты запирания. В Руководстве [3] они называются «current 

technologies – современные технологии». Для элемента SOLID185 доступны 

следующие возможности. 

Метод  -bar ( ̅ или «  с чертой») представляет собой выборочное со-

кращенное интегрирование. При его использовании объемная деформация ап-

проксимируется полиномом степени на единицу меньше, чем степень полинома 

для аппроксимации деформаций. Если для деформаций используется полином 

первой степени, то для объемной деформации – нулевой. Распределение объ-

емной деформации просто заменяется ее средним значением. Отсюда и назва-

ние – черта означает осреднение по объему элемента, а буквой « » традицион-

но обозначают матрицу связи деформаций и перемещений. При построении 

элемента применяется полное интегрирование. Этот метод включен «по умол-

чанию».  

file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SHELL181.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SHELL281.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLSH190.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_PLANE182.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_PLANE183.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLID185.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLID186.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLID187.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLID185.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLID185.html
file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLID185.html

Метод  -bar помогает предотвратить объемное запирание для почти не-

сжимаемых материалов, но не может предотвратить сдвиговое запирание в за-

дачах с преобладанием изгиба. 

В задачах с преобладанием изгиба может оказаться полезной еще одна 

технология элемента SOLID185 – улучшенное представление деформаций (en-

hanced strain formulation). Метод состоит во введении внутренних степеней 

свободы для преодоления сдвигового и объемного запираний. Добавление и 

обработка внутренних неизвестных происходит автоматически без участия 

пользователя. Этот метод предотвращает сдвиговое запирание в задачах с пре-

обладанием изгиба и объемное запирание в случаях с почти несжимаемыми ма-

териалами. 

Выбор технологии построения конечного элемента осуществляется с по-

мощью ключевой опции KEYOPT(2). Значение 0 соответствует полной схеме 

интегрирования с  -bar методом, а значение 2 – методу улучшенного представ-

ления деформаций. 

Все сказанное выше относилось к элементам, построенным на основе 

принципа Лагранжа, которые в меню обозначаются как «Pure displacement». 

Кроме этого, в ANSYS имеется возможность использовать элементы на основе 

так называемой смешанной u-P формулировке – варианте принципа Рейснера 

(«Mixed u-P»). В этом случае в элементах возникают новые неизвестные – зна-

чения гидростатического (или среднего) давления. Мы не будем здесь рассмат-

ривать теоретические основы смешанной формулировки. Желающие могут са-

мостоятельно изучить это вопрос по источникам [5,6]. Элементы на основе 

смешанной u-P формулировки можно применять как к расчету почти несжима-

емых материалов, так и к расчету полностью несжимаемых материалов.  

Использование смешанных (еще говорят – гибридных) конечных элемен-

тов осуществляется с помощью ключевой опции KEYOPT(6)=1. Все отмечен-

ные выше современные технологии, применимые к элементам «Pure 

displacement», применимы также и к элементам «Mixed u-P». 

При выборе конечного элемента для расчетов можно следовать следую-

щему правилу. 

 Для материалов с умеренной сжимаемостью рекомендуется использо-

вать элементы «Pure displacement» с применением современных техноло-

гий; при возникновении затруднений в сходимости переходить к смешан-

ным элементам на основе u-P формулировки. 

 Для материалов с очень слабой сжимаемостью или полностью несжи-

маемых рекомендуется использовать элементы по современной техноло-

гии на основе смешанной u-P формулировки или SOLID285. 

В следующем разделе рассмотрены примеры решения задач для гипе-

рупругих материалов ANSYS. 

file:///C:/Program%20Files/ANSYS%20Inc/v172/commonfiles/help/en-us/help/ans_elem/Hlp_E_SOLID185.html

ПРИМЕРЫ РЕШЕНИЯ ЗАДАЧ ДЛЯ 

ГИПЕРУПРУГИХ МАТЕРИАЛОВ ANSYS 

Растяжение пластины из гиперупругого материала 

Нормальная сжимаемость материала 

Рассмотрим следующую задачу о жёстком растяжении квадратной пла-

стины в условиях плоской деформации. Размеры пластины:           . По 

вертикальным боковым граням заданы перемещения      . Свойства материа-

ла пластины: модуль сдвига     МПа, модуль объёмного сжатия     МПа. 

Задача трёхмерная, пластина имеет три плоскости симметрии, поэтому 

при решении рассматривается     часть полосы с заданием соответствующих 

условий симметрии. Задача решается на сетке         конечных элементов. 

Сравним решения задачи для двух моделей гиперупругих материалов: 

модели неогукова материала и двух константной модели материала Муни–

Ривлина. Необходимый для задания обеих моделей поведения материала, мате-

риальный параметр несжимаемости   определяется соотношением     ⁄  и 

для рассматриваемой задачи         
  . Для двухконстантного материала 

Муни–Ривлина материальные константы    
 ,    

  возьмём равными    
      , 

   
       в соответствии с условием    (   

     
 ). 

Результаты решения задачи для неогукова материала приведены на рис. 3. 

На рис. 3а – распределение перемещения   , на рис. 3б – распределение напря-

жения   . Пунктиром обозначена недеформированная расчетная область. 

Аналогичные результаты решения задачи для материала Муни–Ривлина 

приведены на рис. 4: на рис. 4а – распределение перемещения   , на рис. 4б – 

распределение напряжения   , пунктиром обозначена недеформированная рас-

четная область. 

Сравнивая полученные решения можно утверждать, что качественно ре-

зультаты мало отличаются друг от друга. Количественно отклонение решений 

друг от друга по перемещениям составляет           , по напряжениям – 

           . 

Качественное совпадение результатов обусловлено достаточно большой 

сжимаемостью гиперупругого материала для обеих моделей поведения (и не-

огука, и Муни–Ривлина): константы материалов соответствуют коэффициенту 

Пуассона   (     ) ( (    ))⁄         . 

Задание 

Попробуйте повторить решение задачи самостоятельно. Это не должно за-

нять много времени. Командный файл для решения задачи приведен в прило-

жении 1. 

Слабая сжимаемость материала 

Рассмотрим ту же задачу о жёстком растяжении квадратной пластины, но 

в условиях плоского напряжённого состояния и со свойствами материала Му-

ни–Ривлина (2-х константная модель) использованными в работе [7]:     
     МПа ,          МПа и        МПа  (        МПа  ). Эти данные 

соответствуют начальному значению модуля сдвига        МПа и коэффи-

циента Пуассона          . 

Аналогично предыдущему, трёхмерная задача имеет три плоскости сим-

метрии, поэтому рассматривается 1/8 часть пластины с заданием соответству-

ющих условий симметрии. Задача решается на сетке         конечных 

элементов. 

Результаты решения задачи приведены на рис. 5. На рис. 5а – распределе-

ние перемещения   , на рис. 5б – распределение напряжения   . Пунктиром 

обозначена недеформированная расчетная область. 

Сравним полученное решение с решением, в котором использована мо-

дель неогукова материала. Для этого модуль сдвига зададим равным удвоенной 

сумме параметров Муни–Ривлина. Остальные условия оставим без изменений. 

Аналогичные результаты решения для неогуковой модели материала показано 

на рис. 6. 

Видно, что решение в этом случае значительно отличается от решения в 

первом случае. Иная форма свободного края пластины и значение поперечного 

сужения (       против        в первом случае). Если для материала Муни–

Ривлина в центре пластины имеем однородное напряженное состояние, для не-

огукова материала – неоднородное распределение растягивающих напряжений 

в центре. 

В обоих случаях ANSYS успешно справился с задачей, при этом опция 

автоматической коррекции величины шага нагружения была включена. 

Попробуем решить задачу с использованием смешанных конечных эле-

ментов. Исправим для этого в командном файле значение ключевой опции 

KEYOPT(6)=1. Полученное решение практически совпадает с решением, по-

лученным с помощью элементов, построенных на основе принципа Лагранжа. 

Отличие для материала Муни–Ривлина по напряжениям будет в четвертом зна-

ке после запятой, по перемещениям – в пятом. Но разница в решениях все же 

есть. При решении с помощью элементов в перемещениях начальное значение 

величины шага     для успешного начала вычислений было уменьшено до ве-

личины        , то есть в      раз; на первом шаге потребовалось    итераций 

метода Ньютона. Всего решение было получено за    шагов по нагрузке, а об-

щее количество итераций метода Ньютона равняется    . При решении с ис-

пользованием смешанных элементов начальное значение шага по нагрузке 

осталось неизменным, решение на первом шаге получено за три итерации, все-

го потребовалось   шагов и    итераций по равновесию. Если считать затраты 

на одну итерацию примерно одинаковыми, решение с использованием  смешан- 

ных элементов оказывается в    раз экономичнее решения с использованием 

элементов на основе принципа Лагранжа. Однако, нужно отметить, что в дан-

ном случае мы используем материал со слабой сжимаемостью. 

Задания 

Выполните расчеты растяжения пластины для различных соотношений 

коэффициентов 0110,CC   при условии сохранения начального значения модуля 

сдвига. Постройте графики зависимости формы свободного края. Постройте 

графики распределения нормальных напряжений на оси симметрии. Определи-

те, зависит ли скорость сходимости от соотношения констант материала. 

Определите предельное удлинение пластины, для которого выполняется 

сходимость решения при использовании элементов различного типа.  

Самостоятельно постройте модель пластины с центральным круговым 

отверстием. Выясните зависимость свободного края и формы отверстия от 

соотношения констант материала. 

Обжатие арочного амортизатора 

Существует большое количество технических областей, где используются 

различные опоры, подушки, амортизаторы из гиперупругих материалов. Одно 

из таких устройств – арочный амортизатор [8], изображенный на рис. 5а. 

  
а. Внешний вид б. Расчетная область 

Рис. 5. Арочный амортизатор 

Хотя на рис. 5а показано, что амортизатор нагружен погонной силой, мы 

будем прикладывать заданные перемещения. Во-первых, мы не знаем диапазон 

рабочих нагрузок этого элемента. Во-вторых, именно кинематическая нагрузка 

позволит выяснить некоторые его интересные свойства. Основным размером, 

определяющим геометрию амортизатора, являются высота  , ширина основа-

ния   и угол  . Ширина основания   определяется как доля  . Также через   

определяется и максимальная величина обжатия  . Верхние и нижние горизон-

тальные опоры жестко приклеены или приварены к металлическим частям, по-

этому их можно считать либо жестко закрепленными, либо движущимися как 

жесткое целое. Амортизатор имеет плоскость симметрии. В первом приближе-

нии можно считать, что выполняются условия плоской деформации. Поскольку 

мы планируем для решения 3D конечный элемент, создадим геометрическую 

модель толщиной в «один конечный элемент». В итоге получим расчетную мо-

дель, как показано на рис. 5б. Обсудим задание граничных условий. На лицевой 

и обратной ей поверхностях зададим условия симметрии. Они обеспечат вы-

полнение условий плоской деформации. Нижнее основание жестко закреплено. 

На верхнем основании также можно задать нулевые перемещения по оси   и 

отрицательные перемещения по оси  . 

Нормальная сжимаемость материала 

Сначала проведём сравнение решений для двух моделей поведения гипе-

рупругого материала: модели неогукова материала и двухконстантной модели 

материала Муни–Ривлина, – для свойств с модулем сдвига     МПа и моду-

лем объёмного сжатия     МПа  (соответствуют коэффициенту Пуассона 

         ). Для двухконстантного материала Муни–Ривлина    
      , 

   
       аналогично задаче о жёстком растяжении квадратной пластины. 

Результаты решения задачи для неогукова материала приведены на рис. 7. 

На рис. 7а – распределение перемещения   ,  на рис. 7б –  распределение  экви- 
 

  

а. Перемещение   
  б. Эквивалентное напряжение     

  

Рис. 7. Материал неогука 

  

а. Перемещение   
  б. Эквивалентное напряжение     

  

Рис. 8. Материал Муни–Ривлина 

валентных по Мизесу напряжений     
 . Пунктиром обозначена недеформиро-

ванная расчетная область. Аналогично на рис. 8 приведены результаты решения 

задачи для материала Муни–Ривлина: на рис. 8а – распределение перемещения 

  , на рис. 8б – распределение эквивалентных по Мизесу напряжений     
 . 

Сравнивая полученные решения можно утверждать, что качественно ре-

зультаты мало отличаются друг от друга, количественное различие довольно 

сущестенное: отклонение решений друг от друга по эквилентному по Мизесу 

напряжению –            . Выбор модели гиперупругого материала суще-

ственно влияет на результаты поведения конструкции. 

Задание 

Попробуйте повторить решение задачи самостоятельно. Это не очень слож-

но: командный файл для решения задачи, которым можно воспользоваться, 

приведен в приложении 2. 

Слабая сжимаемость материала 

На основе решения задачи обжатия амортизатора желательно получить 

упругую характеристику амортизатора – кривую, связывающую силу реакции с 

величиной вертикального смещения. Такая зависимость будет весьма полезна 

при проектировании конструкций и устройств с использованием данного амор-

тизатора. Получить величину перемещения верхней границы как функцию 

«времени» не составит труда с использованием временного постпроцессора 

Time-History Postprocessor (POST26). Равнодействующую силу реакции можно 

вычислить как сумму узловых реакций в узлах верхней границы. Чтобы авто-

матизировать вычисление равнодействующей, можно применить простой при-

ем. В ANSYS имеется возможность задавать уравнения связи на узловые неиз-

вестные. В частности, c помощью команды CP можно задать условие, что все 

неизвестные (узловые перемещения) заданного набора узлов имеют одинако-

вые значения. Если нужно задать это значение, это можно сделать, задав его в 

узле, имеющем минимальный в этом наборе номер. Тогда реакция в этом узле 

будет равна равнодействующей реакций всех узлов этого набора. Получить но-

мер минимального в наборе узла можно с помощью команды *GET. Информа-

цию по командам можно получить в [3]. 

Свойства материала будем использовать те же, что и в примере о растя-

жении квадратной пластины при слабой сжимаемости материала Муни–

Ривлина. Используем конечные элементы на основе смешанной формулировки. 

Командный файл решения задачи приведен в приложении 2. 

Приведем результаты решения задачи: на рис. 9 показаны эквивалентные 

напряжения по Мизесу при статическом обжатии арочного амортизатора с ма-

териалом Муни-Ривлина при условии плоской деформации в различные момен-

ты времени. На рис. 9а показано начало процесса. Боковые стороны сечения 

амортизатора практически прямые, но уже видны зоны, в которых концентри-

руются напряжения, на противоположных сторонах стенки возникают зоны 

разгрузки. По мере развития процесса обжатия (рис. 9б) стороны сечения при-

обретают характерную обратную S-образную форму. Зоны концентрации и раз-

грузки так же присутствуют. Изображение на рис. 9в соответствует моменту, 

когда реактивная сила (рис. 10), возникающая при обжатии, максимальна. На 

последнем рисунке (рис. 9г) амортизатор уже практически сложился, реактив-

ная сила уменьшается, скоро начнется контактное взаимодействие стенок самих 

с собой. 

  
а. Условное время 0.07 б. Условное время 0.38 

  
в. Условное время 0.78 г. Условное время 1.0 

Рис. 9. Эквивалентные напряжения по Мизесу  

при статическом обжатии арочного амортизатора 

На рис. 10 показана зависимость равнодействующей силы реакции   от 

вертикального перемещения   при статическом обжатии арочного амортизато-

ра с материалом Муни-Ривлина при условии плоской деформации. На графике 

можно наблюдать критическую точку. Состояния равновесия за этой точкой 

являются неустойчивыми. Они существуют только при кинематическом нагру-

жении, когда мы фиксируем заданные перемещения обжатия. 

Таким образом, мы определили предельную нагрузку, которую может 

выдержать амортизатор. По графику эта нагрузка составляет      . Здесь нуж-

но сделать оговорку. Мы использовали модель плоской деформации и опреде-

лили предельную нагрузку, которая приходится на толщину одного конечного 

элемента, которая в данном расчете составляет     . Следовательно, на      

будет приходиться     . Теперь, зная размер реального амортизатора, можно 

определить для него предельную нагрузку. 

 
Рис. 10. Зависимость реакции  

от вертикального перемещения 

В приложении 3 приведен командный файл для решения трехмерной за-

дачи обжатия арочного амортизатора. 

Задания 

Выполните расчеты обжатия амортизатора для различных соотноше-

ний констант материала в условиях плоской деформации. 

Постройте графики зависимости предельной нагрузки в зависимости от 

угла  . Постройте графики распределения нормальных и касательных напря-

жений на горизонтальных сечениях        ⁄    ⁄     ⁄    . Определите, 

зависит ли скорость сходимости от соотношения констант материала. 

Самостоятельно решите задачу обжатия арочного амортизатора в 

трехмерной постановке (Приложение 3). Сравните величину реакций для слу-

чаев плоской деформации и трехмерного. Объясните причины получения раз-

личных результатов. 

Постройте двумерную модель с использованием элементов PLANE182. 

Дополните модель условиями самоконтакта сторон. Под самоконтактом по-

нимаем контакт одной части стороны сечения с другой ее частью. Проведите 

численные исследования полученной модели с условиями самоконтакта. Воз-

можны ли устойчивые положения равновесия для такой модели? 

Раздувание квадратной мембраны 
равномерным давлением 

Нормальная сжимаемость материала 

Рассматривается задача о раздувании квадратной мембраны       
     равномерным давлением           , приложенным по нижней грани. В 

силу симметрии при численном решении моделируется только одна четверть 

мембраны (первый квадрант в осях    ). На внешних гранях задаётся отсут-

ствие вертикального смещения   
 , на плоскостях симметрии – условия симмет-

рии: равенство нулю нормальных перемещений и касательных усилий. Исполь-

зуется сетка         восьмиузловых конечных элементов SOLID185 c тех-

нологией выборочного сокращенного интегрирования (B-bar стабилизация). 
Аналогично двум предыдущим примерам проведём сравнение решений 

для двух моделей поведения гиперупругого материала: модели неогукова мате-

риала и двухконстантной модели материала Муни–Ривлина, – для свойств с 

модулем сдвига     МПа и модулем объёмного сжатия     МПа (соответ-

ствуют коэффициенту Пуассона          ). Для двухконстантного материа-

ла Муни–Ривлина –    
      ,    

      . 

На рисунках 11-12 приведены результаты решения для неогукова матери-

ала: соответственно на рис. 11 показано распределение вертикальных переме-

щений (  
 ), на рис. 12 – распределение эквивалентных по Мизесу напряжений 

(    
 ). 

 

  

а. Вид сверху б. Аксонометрический вид 

Рис. 11.Перемещение   
  

  

а. Вид сверху б. Аксонометрический вид 

Рис. 12. Эквивалентное (по Мизесу) напряжение     

а. Вид сверху б. Аксонометрический вид 

Рис. 13.Перемещение   
  

  

а. Вид сверху б. Аксонометрический вид 

Рис. 14. Эквивалентное (по Мизесу) напряжение     
  

 

На рисунках 13-14 приведены результаты решения для материала Муни–

Ривлина: на рис. 13 – распределение вертикальных перемещений (   
 ), на 

рис. 14 – распределение эквивалентных по Мизесу напряжений (    
 ). 

В обоих случаях при раздувании мембрана приобретает куполообразную фор-

му. Максимальные эквивалентные напряжения возникают в центре мембраны. Изме-

няется деформированная форма мембраны в плане, наблюдается изменение размеров 

мембраны около середин сторон, внешние стороны мембраны искривляются. 

Аналогично предыдущим примерам для обеих моделей решения каче-

ственно похожи, количественно существенно различаются. 

Задания 

Попробуйте повторить решение задачи самостоятельно. Это не очень слож-

но: командный файл для решения задачи, которым можно воспользоваться, 

приведен в приложении 4. 

Слабая сжимаемость материала 

Рассмотрим ту же задачу о раздувании квадратной мембраны, но со свой-

ствами материала Муни–Ривлина (2-х константная модель):          МПа, 

         МПаи        МПа (        МПа  ). Эти данные соответству-

ют начальному значению модуля сдвига        МПа и коэффициента Пуас-

сона          . 
 

  

а. Вид сверху б. Аксонометрический вид 

Рис. 15.Перемещение  
   

  

а. Вид сверху б. Аксонометрический вид 

Рис. 16. Эквивалентное (по Мизесу) напряжение     
  

 

На рисунках 15-16 приведены результаты решения аналогичные решени-

ям с нормальной сжимаемостью: на рис. 15 – распределение вертикальных пе-

ремещений (  
 ), на рис. 16 – распределение эквивалентных по Мизесу напря-

жений (    
 ). 

Задания 

Выполните расчеты расчёты раздувания квадратной мембраны для раз-

личных соотношений констант материала   ⁄ . Определите, зависит ли ско-

рость сходимости от соотношения констант материала. 

Решите задачу для неогукова материала и таких же соотношений кон-

стант материала   ⁄ . Определите зависимость различия решений от соот-

ношения   ⁄ . 

ПРИЛОЖЕНИЕ 1 

Командный файл решения задачи растяжения пластины 

!*   

/NOPR    

KEYW,PR_SET,1   ! тип решаемой задачи 

KEYW,PR_STRUC,1  

KEYW,PR_THERM,0  

KEYW,PR_FLUID,0  

KEYW,PR_ELMAG,0  

KEYW,MAGNOD,0    

KEYW,MAGEDG,0    

KEYW,MAGHFE,0    

KEYW,MAGELC,0    

KEYW,PR_MULTI,0  

KEYW,PR_CFD,0    

/GO  

!*   

a=10      ! размер расчетной области (половина ширины пластины) 

h=1       ! половина толщины пластины 

n=16      ! число конечных элементов на стороне пластины (половина) 

m=1       ! число конечных элементов по толщине пластины (половина) 

C10=0.31  ! Параметры Муни-Ривлина (МПа или Н/мм2) 

C01=0.19 

G  = 2*(C10+C01)  ! модуль сдвига 

K  = 2            ! объёмный модуль 

v  = (3*K-2*G)/(2*(3*K+G))  ! коэффициент Пуассона 

d  = 2/K  ! параметр несжимаемости 

!*   

/PREP7   

!*   

ET,1,SOLID185 ! выбор типа конечного элемента 

KEYOPT,1,2,0  ! технология КЭ - полное интегрирование + метод Bbar 

KEYOPT,1,6,0  ! элемент на основе принципа Лагранжа (в перемещениях) 

!KEYOPT,1,6,1 ! элемент на основе смешанной формулировки (u-P) 

!*   

! Здесь можно активировать модель неогука, при этом  

! деактивировать модель Муни-Ривлина 

!TB,HYPE,1,1,2,NEO  ! модель неогука (деактивирована) 

!TBTEMP,0 

!TBDATA,,G,d,,,,  

!*   

TB,HYPE,1,1,2,MOON  ! модель Муни-Ривлина (активирована) 

TBTEMP,0 

TBDATA,,C10,C01,d,,,  

!*   

BLOCK,,a,,a,,h,  ! геометрическая модель 

!*   

! Разбивка линий для задания густоты сетки 

LESIZE, 5, , ,n, , , , ,0  

LESIZE, 8, , ,n, , , , ,0  

LESIZE, 9, , ,m, , , , ,0  

!*   

MSHAPE,0,3D ! выбор формы элемента и  

MSHKEY,1  ! типа сетка - mapped - регулярная 

VMESH, 1  ! построение сетки для объёма 1 

!*   

FINISH   

!*   

/SOL 

ANTYPE,0  ! тип анализа - квазистатика 

NLGEOM,1  ! учет геометрической нелинейности 

NSUBST,10,0,0  ! устанавливаем 10 подшагов - приращений приложения 
!                    нагрузки (автоматический выбор подшага 
!                    присутствует) 

!OUTRES,ERASE 

OUTRES,ALL,ALL  ! определение вывода информации 

TIME,1          ! условное время в конце шага 

!*   

DA,1,SYMM  ! условие симметрии на 4 гранях 

DA,2,SYMM  ! обеспечивает плоскую деформацию, для плоского 
!               напряжённого состояния – убрать 

DA,3,SYMM 

DA,5,SYMM 

!*   

DA,6,UX,2*a  ! задание кинематической нагрузки на боковой грани 

DA,6,UY,0 

!*   

solve  ! решить задачу на текущем шаге нагружения 

!*   

FINISH   

ПРИЛОЖЕНИЕ 2 

Командный файл решения задачи обжатия  

арочного амортизатора в условиях плоской деформации 

!*   

/NOPR    

KEYW,PR_SET,1   ! тип решаемой задачи 

KEYW,PR_STRUC,1  

KEYW,PR_THERM,0  

KEYW,PR_FLUID,0  

KEYW,PR_ELMAG,0  

KEYW,MAGNOD,0    

KEYW,MAGEDG,0    

KEYW,MAGHFE,0    

KEYW,MAGELC,0    

KEYW,PR_MULTI,0  

KEYW,PR_CFD,0    

/GO  

!*   

Pi=acos(-1)  ! число Pi 

!*   

H=100        ! высота амортизатора 

b=0.4*H      ! ширина опоры 

delta=0.4*H  ! величина обжатия 

alpha = 60    ! угол при основании в градусах 

!*   

m=10         ! число элементов на основании 

!*   

C10=0.31     ! параметры Муни-Ривлина 

C01=0.19 

G  = 2*(C10+C01) ! модуль сдвига 

K  = 2        ! Объемные модуль 

v=(3*K-2*G)/(2*(3*K+G)) ! коэффициент Пуассона 

d  = 2/K         ! параметр несжимаемости 

!*   

al=alpha/180*Pi ! угол при основании в радианах 

!*   

/PREP7   

!*   

ET,1,SOLID185  ! выбор конечного элемента 

!*   

KEYOPT,1,2,0  ! технология КЭ - полное интегрирование + метод Bbar 

!KEYOPT,1,6,0 ! элемент на основе принципа Лагранжа (в перемещениях) 

KEYOPT,1,6,1 ! элемент на основе смешанной формулировки (u-P) 

!*   

!TB,HYPE,1,1,2,NEO  ! можно активировать модель неогука 

!TBTEMP,0 

!TBDATA,,G,d,,,,  

!*   

TB,HYPE,1,1,2,MOON  ! модель Муни-Ривлина 

TBTEMP,0 

TBDATA,,C10,C01,d,,,  

!*   

t=b/m  ! размер конечного элемента 

!*   

! опорные точки 

k,, 

k,,b, 

k,,H/tan(al)+b,H 

k,,H/tan(al),H 

!*   

a,1,2,3,4  ! фронтальная поверхность 

!*   

VOFFST,1,t, ,  ! "выдавливаем" объем толщиной по размеру КЭ 

!*   

ESIZE,t,0,  ! размер КЭ 

!*   

MSHAPE,0,3D  

MSHKEY,1 

VMESH, 1  ! построение сетки 

!*   

DA, 1,SYMM  ! задаем плоскую деформацию 

DA, 2,SYMM 

!*   

DA, 3,UX,  ! условия заделки на нижней опоре амортизатора 

DA, 3,UY, 

!*   

! Связывание узлов 

NSEL,S,LOC,Y,H           ! выбираем все узлы на нижнем основании 

CM,UpperNodes,NODE       ! создаем компонент (множество) UpperNodes с 
!                                    узлами на верхнем основании 

*GET,upnd,NODE,,NUM,MIN  ! upnd - меньший номер узла в выбранном 
!                                 множестве   

!*   

CP,1,UX,ALL  !все выбранные узлы имеют одинаковое перемещение UX 

CP,2,UY,ALL  !все выбранные узлы имеют одинаковое перемещение UY 

!*   

ALLSEL,ALL    ! активны все узлы 

!*   

D, upnd,UX,        ! граничное условие по X на верхней опоре 

D, upnd,UY,-delta  ! граничное условие по Y на верхней опоре 

!*   

FINISH   

!*   

/SOL 

!*   

ANTYPE,0 

NLGEOM,1 

OUTRES,ERASE 

OUTRES,ALL,ALL 

!AUTOTS,0 

TIME,1   

DELTIM,0.02,0,0.1 

AUTOTS,1 

!*   

solve 

!*   

FINISH   

!*   

/POST1  ! постпроцессор 

!*   

/EDGE,1,1,45  ! отображаем ребра конечных элементов при выводе 
!                   результатов 

/GLINE,1,0   

!/REPLOT 

ПРИЛОЖЕНИЕ 3 

Командный файл решения задачи  

обжатия арочного амортизатора в трехмерной постановке 

!*   

/NOPR    

KEYW,PR_SET,1   ! тип решаемой задачи 

KEYW,PR_STRUC,1  

KEYW,PR_THERM,0  

KEYW,PR_FLUID,0  

KEYW,PR_ELMAG,0  

KEYW,MAGNOD,0    

KEYW,MAGEDG,0    

KEYW,MAGHFE,0    

KEYW,MAGELC,0    

KEYW,PR_MULTI,0  

KEYW,PR_CFD,0    

/GO  

!*   

Pi=acos(-1)  ! число Pi 

!*   

H=100        ! высота амортизатора 

S=80         ! половина длины амортизатора 

b=0.4*H      ! ширина опоры 

delta=0.4*H  ! величина обжатия 

alpha = 60    ! угол при основании в градусах 

!*   

m=10         ! число элементов на основании 

!*   

C10=0.36     ! параметры Муни-Ривлина 

C01=0.22 

G  = 2*(C10+C01) ! модуль сдвига 

K  = 2000        ! объемный модуль 

v=(3*K-2*G)/(2*(3*K+G)) ! коэффициент Пуассона 

d  = 2/K         ! параметр несжимаемости 

!*   

al=alpha/180*Pi ! угол при основании в радианах 

!*   

/PREP7   

!*   

ET,1,SOLID185  ! выбор конечного элемента 

!*   

KEYOPT,1,2,0  ! технология КЭ - полное интегрирование + метод Bbar 

!KEYOPT,1,6,0 ! элемент на основе принципа Лагранжа (в перемещениях) 

KEYOPT,1,6,1 ! элемент на основе смешанной формулировки (u-P) 

!*   

!TB,HYPE,1,1,2,NEO  ! можно активировать модель неогука 

!TBTEMP,0 

!TBDATA,,G,d,,,,  

!*   

TB,HYPE,1,1,2,MOON  ! модель Муни-Ривлина 

TBTEMP,0 

TBDATA,,C10,C01,d,,,  

!*   

t=b/m  ! размер конечного элемента 

nz=S/t ! количество элементов по длине амортизатора 

!*   

! опорные точки 

k,, 

k,,b, 

k,,H/tan(al)+b,H 

k,,H/tan(al),H 

!*   

a,1,2,3,4  ! фронтальная поверхность 

!*   

VOFFST,1,S, ,  ! "выдавливаем" объем длиной S 

!*   

ESIZE,t,0,  ! размер КЭ 

!*   

MSHAPE,0,3D  

MSHKEY,1 

VMESH, 1  ! построение сетки 

!*   

DA, 1,SYMM  ! задаем условие симметрии 

!*   

DA, 3,UX,  ! условия заделки на нижней опоре амортизатора 

DA, 3,UY, 

DA, 3,UZ, 

!*   

! Связывание узлов 

NSEL,S,LOC,Y,H           ! выбираем все узлы на нижнем основании 

CM,UpperNodes,NODE       ! создаем компонент (множество) UpperNodes с 
!                                    узлами на верхнем основании 

*GET,upnd,NODE,,NUM,MIN  ! upnd - меньший номер узла в выбранном 
!                                 множестве   

!*   

CP,1,UX,ALL  !все выбранные узлы имеют одинаковое перемещение UX 

CP,2,UY,ALL  !все выбранные узлы имеют одинаковое перемещение UY 

CP,3,UZ,ALL  !все выбранные узлы имеют одинаковое перемещение UZ 

!*   

ALLSEL,ALL    ! активны все узлы 

!*   

D, upnd,UX,        ! граничное условие по X на верхней опоре 

D, upnd,UY,-delta  ! граничное условие по Y на верхней опоре 

D, upnd,UZ,        ! граничное условие по Z на верхней опоре 

!*   

FINISH   

!*   

/SOL 

!*   

ANTYPE,0 

NLGEOM,1 

OUTRES,ERASE 

OUTRES,ALL,ALL 

!AUTOTS,0 

TIME,1   

DELTIM,0.02,0,0.1 

AUTOTS,1 

!*   

solve 

!*   

FINISH   

!*   

/POST1  ! постпроцессор 

!*   

/EDGE,1,1,45  ! отображаем ребра конечных элементов при выводе 
!                   результатов 

/GLINE,1,0   

!*   

ПРИЛОЖЕНИЕ 4 

Командный файл решения задачи растяжения пластины 

!*   

/NOPR    

KEYW,PR_SET,1   ! тип решаемой задачи 

KEYW,PR_STRUC,1  

KEYW,PR_THERM,0  

KEYW,PR_FLUID,0  

KEYW,PR_ELMAG,0  

KEYW,MAGNOD,0    

KEYW,MAGEDG,0    

KEYW,MAGHFE,0    

KEYW,MAGELC,0    

KEYW,PR_MULTI,0  

KEYW,PR_CFD,0    

/GO  

!*   

a=10      ! размер расчетной области (половина ширины пластины) 

h=1       ! толщина пластины 

n=16      ! число конечных элементов на стороне пластины (половина) 

m=1       ! число конечных элементов по толщине пластины (половина) 

C10=0.36  ! Параметры Муни-Ривлина (МПа или Н/мм2) 

C01=0.22 

G  = 2*(C10+C01)  ! модуль сдвига 

K  = 2000            ! объёмный модуль 

v  = (3*K-2*G)/(2*(3*K+G))  ! коэффициент Пуассона 

d  = 2/K  ! параметр несжимаемости 

q  = 0.11 ! значение давления 

!*   

/PREP7   

!*   

ET,1,SOLID185 ! выбор типа конечного элемента 

KEYOPT,1,2,0  ! технология КЭ - полное интегрирование + метод Bbar 

KEYOPT,1,6,0  ! элемент на основе принципа Лагранжа (в перемещениях) 

!KEYOPT,1,6,1 ! элемент на основе смешанной формулировки (u-P) 

!*   

! Здесь можно активировать модель неогука, при этом  

! деактивировать модель Муни-Ривлина 

!TB,HYPE,1,1,2,NEO  ! модель неогука (деактивирована) 

!TBTEMP,0 

!TBDATA,,G,d,,,,  

!*   

TB,HYPE,1,1,2,MOON  ! модель Муни-Ривлина (активирована) 

TBTEMP,0 

TBDATA,,C10,C01,d,,,  

!*   

BLOCK,,a,,a,,h,  ! геометрическая модель 

!*   

! Разбивка линий для задания густоты сетки 

LESIZE, 5, , ,n, , , , ,0  

LESIZE, 8, , ,n, , , , ,0  

LESIZE, 9, , ,m, , , , ,0  

!*   

MSHAPE,0,3D ! выбор формы элемента и  

MSHKEY,1  ! типа сетка - mapped - регулярная 

VMESH, 1  ! построение сетки для объёма 1 

!*   

FINISH   

!*   

/SOL 

ANTYPE,0  ! тип анализа - квазистатика 

NLGEOM,1  ! учет геометрической нелинейности 

OUTRES,ERASE 

OUTRES,ALL,ALL  ! определение вывода информации 

TIME,1          ! условное время в конце шага 

!*   

DELTIM,0.001,0.0001,0.1 ! начинаем с мелких подшагов (большое 
!                               искажение элементов) 

AUTOTS,1 

!*   

DA,3,SYMM ! условия симметрии 

DA,5,SYMM 

!*   

DA,4,UZ, ! закрепление боковых граней 

DA,6,UZ,  

!*   

SFA,1,1,PRES,q ! давление на грань 1 

!*   

SOLVE   ! решить задачу на текущем шаге нагружения 

!*   

!FINISH   

!*   

/EDGE,1,1,45  ! отображаем ребра конечных элементов при выводе 
!                   результатов 

/GLINE,1,0   

